
Abstract

We introduce the package allMarkovBases for Macaulay2, which is used to compute all
minimal Markov bases of a given toric ideal. The package builds on functionality of 4ti2 by
producing the fiber graph of the toric ideal. The package uses this graph to compute other
properties of the toric ideal such as its indispensable set of binomials as well as its universal
Markov basis.

1 Introduction

Let k be a field and fix a matrix A ∈ Zd×n such that ker(A)∩Nn = {0}, which we call a configuration
matrix. The toric ideal IA ⊆ k[x1, . . . , xn] is the ideal generated by the binomials xu − xv where
u, v ∈ Nn such that u− v ∈ ker(A). Note that here we use multi-index notation, so xu denotes the
monomial xu1

1 xu2
2 · · ·xun

n . For an introduction to toric ideals, see [Stu96]. A (minimal) Markov basis
of A is an (inclusion-minimal) set of generators M of IA and we identify M with its set of binomial
exponents {u− v : xu − xv ∈ M}. The seminal paper of Diaconis and Sturmfels [DS98] provides a
connection between Markov bases of toric ideals and sampling from conditional distributions. The
research landscape of Algebraic Statistics has since seen a great deal of development, particularly
in the study of Markov bases. For an introduction to Markov bases in Statistics we refer to [Kos20]
and [FAHP24], and for a general introduction to Algebraic Statistics we refer to [Sul23].

The current version of Macaulay2 [GS] is shipped with 4ti2 [HHK+], which is a powerful library
for performing computations with toric ideals, polyhedra, and lattices. Users of Macaulay2 may
be familiar with its interfacing package FourTiTwo, which provides the function toricMarkov that
computes a single Markov basis from a given matrix. However, this package does not provide a
way to study all possible minimal Markov bases of a given matrix A. So we introduce the package
allMarkovBases, which comes with a function markovBases that returns all minimal Markov bases.
We do this by computing the fiber graphs of A, which is a sequence of graphs whose connectivity can
be used to determine the Markov bases of A. See Section 2.1. Our package exploits the fiber graph
to compute important families of binomials of the toric ideal, which cannot be computed directly
with FourTiTwo. This includes the indispensable set S(A) and universal Markov basis U(A), which
are, respectively, the intersection and union of all minimal Markov bases. These families of elements
have alternative combinatorial descriptions and prominently feature in the study of Markov bases.
See, for example, [CTV14, KTV23, CK24, CTV17]. If the number of Markov bases is very large,
then our package allows for the uniform sampling from the set of Markov bases.

1.1 Outline with examples

The matrix A =
(
7 8 9 10

)
has four minimal Markov bases, which are computed as follows.

i1 : needsPackage "allMarkovBases ";

i2 : A = matrix "7,8,9,10";

i3 : countMarkov A

o3 = 4

i4 : markovBases A

o4 = {| -1 2 -1 0 |, | -1 2 -1 0 |, | -1 2 -1 0 |, | -1 2 -1 0 |}

| -1 1 1 -1 | | -1 1 1 -1 | | -1 1 1 -1 | | -1 1 1 -1 |

| 0 -1 2 -1 | | 0 -1 2 -1 | | 0 -1 2 -1 | | 0 -1 2 -1 |

| 4 0 -2 -1 | | 4 0 -2 -1 | | 4 -1 0 -2 | | 4 -1 0 -2 |

| 3 1 -1 -2 | | 3 1 -1 -2 | | 3 1 -1 -2 | | 3 1 -1 -2 |

| 3 0 1 -3 | | 2 2 0 -3 | | 3 0 1 -3 | | 2 2 0 -3 |

On the other hand, the matrix A′ =
(
51 52 53 54 55 56

)
has 24300 minimal Markov bases.

It may not be feasible to work with all of them so we can produce random samples as follows.

i5 : A’ = matrix "51 ,52 ,53 ,54 ,55 ,56";

i6 : countMarkov A’

o6 = 24300

i7 : randomMarkov A’

o7 = | 8 4 0 0 0 -11 |

| -1 2 -1 0 0 0 |

| -1 1 1 -1 0 0 |

| -1 0 2 0 -1 0 |

| -1 1 0 1 -1 0 |

| 0 -1 1 1 -1 0 |

| -1 1 0 0 1 -1 |

| 0 0 -1 2 -1 0 |

| 0 -1 0 2 0 -1 |

| 0 0 -1 1 1 -1 |

| 0 0 0 -1 2 -1 |

| 12 0 -1 0 -1 -9 |

| 11 1 0 -1 -1 -9 |

| 11 0 1 -1 0 -10 |

| 9 3 0 0 -1 -10 |

The indispensable set S(A′) and the universal Markov basis U(A′) are computed as follows.
We have abridged the output of toricUniversalMarkov A’ to save space. We note that the
indispensable set and universal Markov bases are read from the fiber graph, so the methods do not
require the computation of all 24300 minimal Markov bases.

i8 : toricIndispensableSet A’

o8 = | -1 2 -1 0 0 0 |

| -1 1 1 -1 0 0 |

| 0 0 -1 1 1 -1 |

| 0 0 0 -1 2 -1 |

i9 : toricUniversalMarkov A’

o9 = | 10 0 2 0 0 -11 |

| 8 4 0 0 0 -11 |

...

| 9 3 0 0 -1 -10 |

| 11 0 0 1 -1 -10 |

33 6

o9 : Matrix ZZ <-- ZZ

An example from Algebraic Statistics is the Segre embedding of a product of projective spaces:

P2 × P2 × P2, whose toric ideal is given by the matrix

A′′ =



1 1 1 1 1 1 1 1 1 1 1 1 1 1 1 1 1 1 1 1 1 1 1 1 1 1 1
1 1 1 1 1 1 1 1 1 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 1 1 1 1 1 1 1 1 1 0 0 0 0 0 0 0 0 0
1 1 1 0 0 0 0 0 0 1 1 1 0 0 0 0 0 0 1 1 1 0 0 0 0 0 0
0 0 0 1 1 1 0 0 0 0 0 0 1 1 1 0 0 0 0 0 0 1 1 1 0 0 0
1 0 0 1 0 0 1 0 0 1 0 0 1 0 0 1 0 0 1 0 0 1 0 0 1 0 0
0 1 0 0 1 0 0 1 0 0 1 0 0 1 0 0 1 0 0 1 0 0 1 0 0 1 0


.

In this case, we were able to compute the number of Markov bases, the indispensable set, and
universal Markov basis in under a second.

i10 : A’’ = matrix {{1,1,1,1,1,1,1,1,1,1,1,1,1,1,1,1,1,1,1,1,1,1,1,1,1,1,1},

{1,1,1,1,1,1,1,1,1,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0},

{0,0,0,0,0,0,0,0,0,1,1,1,1,1,1,1,1,1,0,0,0,0,0,0,0,0,0},

{1,1,1,0,0,0,0,0,0,1,1,1,0,0,0,0,0,0,1,1,1,0,0,0,0,0,0},

{0,0,0,1,1,1,0,0,0,0,0,0,1,1,1,0,0,0,0,0,0,1,1,1,0,0,0},

{1,0,0,1,0,0,1,0,0,1,0,0,1,0,0,1,0,0,1,0,0,1,0,0,1,0,0},

{0,1,0,0,1,0,0,1,0,0,1,0,0,1,0,0,1,0,0,1,0,0,1,0,0,1,0}};

i11 : elapsedTime countMarkov A’’

-- .306693s elapsed

o11 = 324518553658426726783156020576256

i12 : elapsedTime toricIndispensableSet A’’;

-- .00462137s elapsed

81 27

o12 : Matrix ZZ <-- ZZ

i13 : elapsedTime toricUniversalMarkov A’’;

-- .0104826s elapsed

243 27

o13 : Matrix ZZ <-- ZZ

In the above, most of the time is spent computing the fiber graph. Once the graph has been
computed, it is stored to avoid re-computation. The above example is particularly amenable to our
computations because the fiber graphs are all very small.

2 How it works

Throughout this section, we take A to be a d×n configuration matrix. We explain how our functions
work with the simple running example A =

(
1 2 3

)
∈ Z1×3. First, we require a distinguished

Markov basis M , which is computed by FourTiTwo as follows.

i1 : needsPackage "allMarkovBases ";

i2 : A = matrix "1,2,3";

i3 : M = toricMarkov A

o3 = | 2 -1 0 |

| 3 0 -1 |

So the toric ideal IA ⊆ k[x, y, z] is minimally generated by the binomials x2 − y and x3 − z. The
affine semigroup of A is NA := {Ax ∈ Zd : x ∈ Nn} and, for each t ∈ NA, the t-fiber of A is the
subset Ft := {u ∈ Nn : Au = t}. For each binomial xu − xv ∈ IA, we have that u, v ∈ Ft belong to
the same fiber of A, and we define its A-degree degA(x

u− xv) = t. In our running example we have

degA(x
2 − y) = 2 and degA(x

3 − z) = 3.

By [CKT07, Theorems 2.6 and 2.7] the multiset of A-degrees is the same for any minimal Markov
basis. For instance, the binomials x2 − y, xy − z form another minimal Markov basis for IA and

(a) A-degree 2 (b) A-degree 3

Figure 1: Fiber graphs of A for the generating fibers

and their A-degrees are 2 and 3 respectively. We say that t ∈ NA is a generating fiber for A if
there exists an element xu − xv of a minimal Markov basis with A-degree degA(x

u − xv) = t. The
aforementioned results provide a natural procedure to generate all minimal Markov bases using the
fiber graph of A.

2.1 Fiber graphs

Let t ∈ Nd. The fiber graph Gt has vertex set Ft and edge set

E(Gt) = {uv : there exists i ∈ {1, . . . , n} such that ui > 0 and vi > 0}.

Using our package, the list of fiber graphs of the generating fibers of A are computed as follows.

i4 : fiberGraph A

o4 = {Graph{{0, 1, 0} => {}}, Graph{{0, 0, 1} => {} }}

{2, 0, 0} => {} {1, 1, 0} => {{3, 0, 0}}

{3, 0, 0} => {{1, 1, 0}}

A visualisation of these fibers, produced by the package Graphs, is shown in Figure 1. Each
element z ∈ ker(A) is thought of as a move that connects pairs of points u, v ∈ Nn whenever
u = v + z. In our example, the move (2,−1, 0) connects (2, 0, 0) and (0, 1, 0) and connects (1, 1, 0)
and (3, 0, 0). The fundamental theorem of Markov bases [DS98, Theorem 3.1] tells us that the set of
Markov bases are in one-to-one correspondence with collections of elements of ker(A) that connect
each fiber of A. Here, a subset M ⊆ ker(A) is said to connect the t-fiber, for some t ∈ NA, if the
graph GM,t on Ft with edges {uv : u−v ∈M} is connected. Similarly, by [CKT07, Theorems 2.6 and
2.7], the minimal Markov bases are in one-to-one correspondence with sets of moves that minimally
connect Gt for each generating fiber t of A.

Remark 2.1. In the literature, the graphs GM,t are often called fiber graphs [HM09]. To avoid
confusion, for us the term fiber graph will always refer to the graph Gt for some t.

In our example, we observe that the 2-fiber in Figure 1a is connected only by the element
(2,−1, 0), up to sign. It follows that the corresponding binomial x2 − y is indispensable as there is
no other way to connect the 2-fiber, thus every Markov basis must contain it. On the other hand,
the 3-fiber in Figure 1b is minimally connected with either the move (1, 1,−1) or (3, 0,−1). These
form the two minimal Markov bases of A. This process is carried out by the function markovBases:

i5 : markovBases A

o5 = {| 2 -1 0 |, | 2 -1 0 |}

| 3 0 -1 | | 1 1 -1 |

2.2 The all-Markov algorithm

Our package computes all Markov bases with an implementation of Algorithm 1. By [CKT07,
Theorems 2.6 and 2.7], this algorithm is correct and returns all minimal Markov bases of A.

Figure 2: Spanning tree on {0, 1, . . . , 5} associated to the Prüfer sequence {0, 0, 2, 4}

Algorithm 1: All minimal Markov bases of a matrix

Input : A ∈ Zd×n a configuration matrix
Output:M the set of minimal Markov bases of A

1: InitialiseM = ∅ ;
2: for t ∈ NA a generating fiber of A do
3: Compute Ct := the set of connected components of the fiber graph Gt ;
4: end

5: T :=
∏

t{T : spanning tree on vertex set Ct}, where the product is taken over all
generating fibers t ∈ NA of A

6: for {Tt : spanning tree on Ct} ∈ T do
7: for each collection {fT,e : T ∈ T , e ∈ E(T )} where fT,e is a choice function on e do
8: M := {fT,e(u)− fT,e(v) : T ∈ T , uv ∈ E(T )} Markov basis of A;
9: M←M∪ {M};

10: end

11: end

Let us give some notes on how the algorithm works. For each fiber graph Gt, with t ∈ NA a
generating fiber of A, we treat its connected components as the vertices of a new graph Ĝt for which
we choose a spanning tree T . Then the union over all generating fibers of the binomials xu − xv,
where u, v are in distinct connected components of Gt which are connected by an edge of T , forms a
minimal Markov basis. Concretely, on line 7, the edge e ∈ E(T ) is a pair of connected components
of Ct, hence e is set of sets of elements of Nn. Suppose that uv ∈ E(T ) is an edge of some T ∈ T ,
then the choice function fT,e produces the binomial xu − xv, which appears in the Markov basis.

The spanning trees are enumerated using the well-known bijection of Prüfer [Prü18]. This bijec-
tion between labelled spanning trees on n vertices and Prüfer sequences (sequences in {0, . . . , n−1}
of length n−2) is completely constructive. We have implemented a helpful function pruferSequence

that takes a Prüfer sequence and returns the edge set of the corresponding spanning tree. For ex-
ample, the spanning tree associated to the sequence {0, 1, 2, 3} is given computed as follows, see
Figure 2.

i6 : pruferSequence {0,0,2,4}

o6 = {set {0, 1}, set {0, 3}, set {0, 2}, set {4, 2}, set {4, 5}}

By knowing only the sizes of connected components of each fiber graph, it is possible to calculate
the number of minimal Markov bases. Let t ∈ NA be a generating fiber of A. Let us assume
Gt has nt connected components of size mt,1,mt,2, . . . ,mt,nt for some positive integers mt,i with
i ∈ {1, 2, . . . , nt}. Then, by a slight modification of Prüfer’s bijection, see [CKT07, Theorem 2.9],
the number of minimal Markov bases of A is given by∏

t

mt,1 ·mt,2 · · ·mt,nt · (mt,1 +mt,2 + · · ·+mt,nt)
nt−2.

In our package, we have implemented this as the function countMarkov, which computes the
number of minimal Markov bases without enumerating all of them.

2.3 Variations on the algorithm

Our package includes variations of Algorithm 1, which allows us to sample from the set of Markov
bases and compute the indispensable set and universal Markov basis.

Random sampling. The function randomMarkov produces a random Markov basis as follows. In
lines 6 and 7 of Algorithm 1, we sample uniformly an element of T and a random collection of
choice functions respectively. This produces a single uniformly distributed minimal Markov basis
of A. The optional argument NumberOfBases specifies how many samples to produces.

i7 : randomMarkov(A, NumberOfBases => 1)

o7 = | 2 -1 0 |

| 3 0 -1 |

Indispensable set and universal Markov basis. The universal Markov basis U(A) of the matrix
A is the union of its minimal Markov bases. Every minimal Markov basis arises from Algorithm 1,
so we may compute U(A) with a slight modification. Specifically, our package provides the function
toricUniversalMarkov, which implements the following modification to the algorithm.

Algorithm 2: Universal Markov basis of a matrix

Input : A ∈ Zd×n a configuration matrix
Output: U(A) the universal Markov basis of A

1: Initialise U(A) = ∅ ;
2: for t ∈ NA a generating fiber of A do
3: Compute Ct := the set of connected components of the fiber graph Gt ;
4: Kt := complete graph on vertex set Ct ;

5: end

6: K :=
⊔

t Kt, where the union is taken over all generating fibers t ∈ NA of A

7: for {u, v} ∈ E(K) do
8: for each choice function fuv on {u, v} do
9: U(A)← U(A) ∪ {fuv(u)− fuv(v)} ;

10: end

11: end

By [CKT07, Corollary 2.10], the indispensable elements can be detected directly from the fiber
graphs. The function toricIndispensableSet determines the indispensable elements by finding
each fiber graph Gt with exactly two vertices that are each singletons. In our running example, the
universal Markov basis and indispensable set are computed as follows.

i8 : toricUniversalMarkov A

o8 = | 2 -1 0 |

| 3 0 -1 |

| 1 1 -1 |

i9 : toricIndispensableSet A

o9 = | 2 -1 0 |

Polynomial output. For each method: markovBases, randomMarkov, toricUniversalMarkov,
toricIndispensableSet, we may specify a polynomial ring. The function returns an ideal whose
generating set, accessible with gens, is the corresponding set of binomials.

i10 : R = QQ[x, y, z];

i11 : markovBases(A, R)

2 3 2

o11 = {ideal (x - y, x - z), ideal (x - y, x*y - z)}

i12 : toricIndispensableSet(A, R)

2

o12 = ideal(x - y)

i13 : gens toricUniversalMarkov(A, R)

o13 = | x2 -y x3 -z xy -z |

